# Maintainer: Sven-Hendrik Haase <sh@lutzhaase.com>
# Contributor: Jameson Pugh <imntreal@gmail.com>

# ALARM: Kevin Mihelich <kevin@archlinuxarm.org>
#  - disable mmx

pkgname=sdl2_gfx
pkgver=1.0.4
pkgrel=4
epoch=1
pkgdesc="SDL graphics drawing primitives and other support functions (Version 2)"
url="http://www.ferzkopp.net/wordpress/2016/01/02/sdl_gfx-sdl2_gfx/"
arch=(x86_64)
license=(zlib)
depends=(sdl2)
source=("https://www.ferzkopp.net/Software/SDL2_gfx/SDL2_gfx-${pkgver}.tar.gz")
sha512sums=('81a100d3c8c3a7c6bd37a23f1290ff10685f8e62fbecd83b0086aae4edc721483e2153cd4219fbd9168f115eea0ea6b25f9be375faf5761f0babdfb1b52fe482')

build() {
  cd SDL2_gfx-$pkgver
  ./configure --prefix=/usr --disable-mmx
  make
}

package() {
  cd SDL2_gfx-$pkgver
  make DESTDIR="$pkgdir" install
  install -Dm644 COPYING "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
